



## Teorema del invariante: corrección de ciclos

**Ejercicio 1.** ★ Consideremos el problema de sumar los elementos de un arreglo y la siguiente implementación en SmallLang, con el invariante del ciclo.

### Especificación

```
proc sumar (in s: seq(Z), out result: Z) {
  Pre {true}
  Post {result =  $\sum_{j=0}^{|s|-1} s[j]$ }
}
```

### Implementación en SmallLang

```
result := 0;
i := 0;
while (i < s.size()) do
  result := result + s[i];
  i := i + 1
endwhile
```

### Invariante de Ciclo

$$I \equiv 0 \leq i \leq |s| \wedge_L result = \sum_{j=0}^{i-1} s[j]$$

- Escribir la precondition y la postcondition del ciclo.
- ¿Qué punto falla en la demostración de corrección si el primer término del invariante se reemplaza por  $0 \leq i < |s|$ ?
- ¿Qué punto falla en la demostración de corrección si el límite superior de la sumatoria  $(i - 1)$  se reemplaza por  $i$ ?
- ¿Qué punto falla en la demostración de corrección si se invierte el orden de las dos instrucciones del cuerpo del ciclo?
- Demostrar formalmente la corrección parcial del ciclo, usando los primeros puntos del teorema del invariante.
- Proponer una función variante y demostrar formalmente la terminación del ciclo, utilizando la función variante.

**Ejercicio 2.** ★ Dadas la especificación y la implementación del problema `sumarParesHastaN`, escribir la precondition y la postcondition del ciclo, y demostrar formalmente su corrección a través del teorema del invariante.

### Especificación

```
proc sumarParesHastaN (in n: Z, out result: Z) {
  Pre {n ≥ 0}
  Post {result =  $\sum_{j=0}^{n-1} (\text{if } j \bmod 2 = 0 \text{ then } j \text{ else } 0 \text{ fi})$ }
}
```

### Implementación en SmallLang

```
result := 0;
i := 0;
while (i < n) do
  result := result + i;
  i := i + 2
endwhile
```

### Invariante de ciclo

$$I \equiv 0 \leq i \leq n + 1 \wedge i \bmod 2 = 0 \wedge result = \sum_{j=0}^{i-1} (\text{if } j \bmod 2 = 0 \text{ then } j \text{ else } 0 \text{ fi})$$

**Ejercicio 3.** Supongamos que se desea implementar la función **exponenciacion**, cuya especificación es la siguiente:

```
proc exponenciacion (in m:  $\mathbb{Z}$ , in n:  $\mathbb{Z}$ , out result:  $\mathbb{Z}$ ) {
  Pre { $n \geq 0 \wedge \neg(m = 0 \wedge n = 0)$ }
  Post { $result = m^n$ }
}
```

Consideremos además el siguiente invariante:  $I \equiv 0 \leq i \leq n \wedge result = m^i$

- Escribir un programa en SmallLang que resuelva este problema, y que incluya un ciclo que tenga a  $I$  como invariante. Demostrar formalmente la corrección de este ciclo.
- La siguiente implementación en SmallLang es trivialmente errónea. ¿Qué punto del teorema del invariante falla en este caso?<sup>1</sup>

```
i := 0;
result := 0;

while( i < m ) do
  result := result * n;
  i := i + 1
endwhile
```

- ¿Qué puede decir de la siguiente implementación en SmallLang? En caso de que sea correcta, proporcione una demostración. En caso de que sea incorrecta, explique qué punto del teorema del invariante falla.

```
i := 0;
result := 1;
while( i < n ) do
  i := i + 1;
  result := result * m
endwhile
```

- ¿Qué puede decir de la siguiente implementación? En caso de que sea incorrecta, ¿se puede reforzar la precondition del problema para que esta especificación pase a ser correcta?

```
i := 2;
result := m*m;

while( i < n ) do
  result := result * m;
  i := i + 1
endwhile
```

**Ejercicio 4. ★** Considere el problema **sumaDivisores**, dado por la siguiente especificación:

```
proc sumaDivisores (in n:  $\mathbb{Z}$ , out result:  $\mathbb{Z}$ ) {
  Pre { $n \geq 1$ }
  Post { $result = \sum_{j=1}^n (\text{if } n \bmod j = 0 \text{ then } j \text{ else } 0 \text{ fi})$ }
}
```

- Escribir un programa en SmallLang que satisfaga la especificación del problema y que contenga exactamente un ciclo.
- El ciclo del programa propuesto, ¿puede ser demostrado mediante el siguiente invariante?

$$I \equiv 1 \leq i \leq n \wedge result = \sum_{j=1}^i (\text{if } n \bmod j = 0 \text{ then } j \text{ else } 0 \text{ fi})$$

Si no puede, ¿qué cambios se le deben hacer al invariante para que se corresponda con el ciclo propuesto?

---

<sup>1</sup>Recordar que para mostrar que una implicación  $A \rightarrow B$  no es cierta alcanza con dar valores de las variables libres que hagan que  $A$  sea verdadero y que  $B$  sea falso. Para mostrar que una tripla de Hoare  $\{P\}S\{Q\}$  no es válida, alcanza con dar valores de las variables que satisfacen  $P$ , y tales que luego de ejecutar  $S$ , el estado final no satisface  $Q$ .

**Ejercicio 5.** Considere la siguiente especificación de la función `sumarPosicionesImpares`.

```
proc sumarPosicionesImpares (in s: seq<Z>, out result: Z) {
  Pre {true}
  Post {result =  $\sum_{i=0}^{|s|-1} (\text{if } i \bmod 2 = 1 \text{ then } s[i] \text{ else } 0 \text{ fi})$ }
}
```

- Implementar un programa en SmallLang para resolver este problema, que incluya exactamente un ciclo con el siguiente invariante:  

$$I \equiv 0 \leq j \leq |s| \wedge_L \text{result} = \sum_{i=0}^{j-1} (\text{if } i \bmod 2 = 1 \text{ then } s[i] \text{ else } 0 \text{ fi})$$
- Demostrar formalmente la corrección del ciclo propuesto.

**Ejercicio 6.** Considere la siguiente especificación e implementación del problema `maximo`.

#### Especificación

```
proc maximo (in s: seq<Z>, out i: Z) {
  Pre {|s| ≥ 1}
  Post {0 ≤ i < |s| ∧L
    (∀j : Z)(0 ≤ j < |s| →L s[j] ≤ s[i])}
}
```

#### Implementación en SmallLang

```
i := 0;
j := 1;
while (j < s.size()) do
  if (s[j] > s[i])
    i := j
  else
    skip
  endif;
  j:=j+1
endwhile
```

- Escribir la precondition y la postcondición del ciclo.
- Demostrar que el ciclo es parcialmente correcto, utilizando el siguiente invariante:

$$I \equiv (0 \leq i < |s| \wedge 1 \leq j \leq |s|) \wedge_L (\forall k : \mathbb{Z})(0 \leq k < j \rightarrow_L s[k] \leq s[i])$$

- Proponer una función variante que permita demostrar que el ciclo termina.

**Ejercicio 7. ★** Considere la siguiente especificación e implementación del problema `copiarSecuencia`.

#### Especificación

```
proc copiarSecuencia (in s: seq<Z>, inout r: seq<Z>) {
  Pre {|s| = |r| ∧ r = r0}
  Post {|s| = |r| ∧L (∀j : Z)(0 ≤ j < |s| →L s[j] = r[j])}
}
```

#### Implementación en SmallLang

```
i := 0;
while (i < s.size()) do
  r[i] := s[i];
  i:=i+1
endwhile
```

- Escribir la precondition y la postcondición del ciclo.
- Proponer un invariante y demostrar que el ciclo es parcialmente correcto.
- Proponer una función variante que permita demostrar que el ciclo termina.

**Ejercicio 8.** Considere la siguiente especificación e implementación del problema `llenarSecuencia`.

#### Especificación

```
proc llenarSecuencia (inout s: seq<Z>, in d: Z, in e: Z) {
  Pre {d ≥ 0 ∧ d < |s| ∧ s = S0}
  Post {|s| = |S0| ∧L
    (∀j : Z)(0 ≤ j < d →L s[j] = S0[j]) ∧
    (∀j : Z)(d ≤ j < |s| →L s[j] = e)}
}
```

#### Implementación en SmallLang

```
i := d;
while (i < s.size()) do
  s[i] := e;
  i:=i+1
endwhile
```

- Escribir la precondition y la postcondición del ciclo.

- b) Proponer un invariante y demostrar que el ciclo es parcialmente correcto.
- c) Proponer una función variante que permita demostrar que el ciclo termina.

**Ejercicio 9.** ★ Sea el siguiente ciclo con su correspondiente precondition y postcondition:

```

while ( i >= length(s) / 2) do
  suma := suma + s[length(s)-1-i];
  i := i - 1
endwhile

```

$$P_c : \{|s| \bmod 2 = 0 \wedge i = |s| - 1 \wedge suma = 0\}$$

$$Q_c : \{|s| \bmod 2 = 0 \wedge i = |s|/2 - 1 \wedge_L suma = \sum_{j=0}^{|s|/2-1} s[j]\}$$

- a) Especificar un invariante de ciclo que permita demostrar que el ciclo cumple la postcondition.
- b) Especificar una función variante que permita demostrar que el ciclo termina.
- c) Demostrar formalmente la corrección y terminación del ciclo usando el Teorema del invariante.

**Ejercicio 10.** Considere la siguiente especificación del problema **reemplazarTodos**.

```

proc reemplazarTodos (inout s: seq(Z), in a: Z, in b: Z) {
  Pre {s = S0}
  Post {|s| = |S0| ∧L
  (∀j:Z)((0 ≤ j < |s| ∧L S0[j] = a) →L s[j] = b) ∧
  (∀j:Z)(d ≤ j < |s| ∧L S0[j] ≠ a) →L s[j] = S0[j])}
}

```

- a) Dar un programa en SmallLang que implemente la especificación dada.
- b) Escribir la precondition y la postcondition del ciclo.
- c) Proponer un invariante y demostrar que el ciclo es parcialmente correcto.
- d) Proponer una función variante que permita demostrar que el ciclo termina.

## Demostración de correctitud: programas completos

**Ejercicio 11.** ★ Demostrar que el siguiente programa es **correcto** respecto a la especificación dada.

### Especificación

```

proc indice (in s: seq(Z), in e: Z, out r: Z) {
  Pre {True}
  Post {r = -1 →
  (∀j:Z)(0 ≤ j < |s| →L s[j] ≠ e)
  ∧
  r ≠ -1 →
  (0 ≤ r < |s| ∧L s[r] = e)}
}

```

### Implementación en SmallLang

```

i := s.size() - 1;
j := -1;
while (i >= 0) do
  if (s[i] = e) then
    j := i
  else
    skip
  endif;
  i := i - 1
endwhile;
r := j;

```

**Ejercicio 12. ★** Demostrar que el siguiente programa es correcto respecto a la especificación dada.

**Especificación**

```
proc existeElemento (in s: seq(Z), in e: Z, out r: Bool) {
  Pre {True}
  Post {r = True ↔
        ((∃k : Z)(0 ≤ k < |s|) ∧L s[k] = e)}
}
```

**Implementación en SmallLang**

```
i := 0;
j := -1;
while (i < s.size()) do
  if (s[i] = e) then
    j := i
  else
    skip
  endif;
  i := i + 1
endwhile;
if (j != -1)
  r := true
else
  r := false
endif
```

**Ejercicio 13.** Demostrar que el siguiente programa es correcto respecto a la especificación dada.

**Especificación**

```
proc esSimetrico (in s: seq(Z), out r: Bool) {
  Pre {True}
  Post {r = True ↔ (∀i : Z)(0 ≤ i < |s| →L
    s[i] = s[|s| - (i + 1)])}
}
```

**Implementación en SmallLang**

```
i := 0;
j := s.size() - 1;
r := true;
while (i < s.size()) do
  if (s[i] != s[j]) then
    r := false
  else
    skip
  endif;
  i := i + 1;
  j := j - 1;
endwhile
```

**Ejercicio 14. ★** Demostrar que el siguiente programa es correcto respecto a la especificación dada.

**Especificación**

```
proc concatenarSecuencias (in a: seq(Z),
in b: seq(Z),
inout r: seq(Z)) {
  Pre {|r| = |a| + |b| ∧ r = R0}
  Post {|r| = |R0| ∧ (∀j : Z)(0 ≤ j < |a| →L r[j] = a[j]) ∧
    (∀j : Z)(0 ≤ j < |b| →L r[j + |a|] = b[j])}
}
```

**Implementación en SmallLang**

```
i := 0;
while (i < a.size()) do
  r[i] := a[i];
  i := i + 1
endwhile;
i := 0;
while (i < b.size()) do
  r[a.size() + i] := b[i];
  i := i + 1
endwhile
```

**Ejercicio 15.** Dar dos programas en SmallLang que satisfagan la siguiente especificación, y demostrar que ambos son correctos.

```
proc buscarPosicionUltimoMaximo (in s: seq(Z), out r: Z) {
  Pre {|s| > 0}
  Post {0 ≤ r < |s| ∧L
    (∀j : Z)(0 ≤ j < r →L s[r] ≥ s[j]) ∧
    (∀j : Z)(r < j < |s| →L s[r] > s[j])}
}
```

## RESOLUCIONES.

### Ejercicio 1

a)

$$Pre : \{result = 0 \wedge i = 0\}$$

$$Post : \{result = \sum_{j=0}^{|s|-1} s[j]\}$$

b) Si le sacas el igual a la segunda desigualdad no te agarra el ultimo elemento la sumatoria.